Dynamic Optimization: CSTR Case Study

A reactor is used to convert a hazardous chemical "*A*" to an acceptable chemical "*B*" in waste stream before entering a nearby lake. This particular reactor is dynamically modeled as a Continuously Stirred Tank Reactor (CSTR) with a simplified kinetic mechanism that describes the conversion of reactant *A* to product *B* with an irreversible and exothermic reaction. It is desired to maintain the temperature at a constant setpoint that maximizes the destruction of *A* (highest possible temperature). You may use the MATLAB, Simulink, and Excel workbook resources to fit models, simulate the system, and perform estimation and control.



Model Predictive Control (Empirical Model)

a) Perform the necessary open loop dynamic modeling studies to determine a first order model ($\tau_p \frac{dT_r}{dt} = -T_r + K_p\, T_c$) which describes the relationship between cooling jacket temperature (MV=T_c) and reactor temperature (CV=T_r). Start at 280K for the cooling jacket temperature (green circled T_c) and step to 300K and down to 260K and back to 300K in a doublet test. Use the file *data.txt* to extract the information for the model.



b) Use the K_p and τ_p from part *a* in a linear model predictive controller (blue circled controller). The files that are downloaded with the problem are with a nonlinear model predictive controller that uses the full first-principles model. Replace this model with a linear MPC controller. Test the set point tracking capability (red circled reactor SP) of this controller by plotting the response of the process to steps in setpoint of the reactor temperature (not cooling jacket temperature) from 300K up to 320K and then down to 280K. Compare to PID control performance. Comment on how the nonlinear behavior of this process impacts your observed set point response performance.



c) Step up the setpoint to achieve a maximum temperature in the reactor without exceeding the maximum allowable temperature of 400K (don't cause a reactor run-away). What is the lowest concentration that can be achieved without exceeding the maximum allowable temperature?

First Principles Modeling Approach

A species balance for A and an energy balance, is used to derive a model of the CSTR response of C_a and T to changes in the inputs $C_{a,i}$, T_i, and T_c.

```
Tc = 270              % Temperature of cooling jacket (K)
q = 100               % Volumetric Flowrate (m^3/sec)
V = 100               % Volume of CSTR (m^3)
ρ = 1000              % Density of A-B Mixture (kg/m^3)
Cp = .239             % Heat capacity of A-B Mixture (J/kg-K)
ΔHr = 5e4             % Heat of reaction for A->B (J/mol)
E/R = 8750            % EoverR = E/R
k0 = 7.2e10           % Pre-exponential factor (1/sec)
UA = 5e4              % Overall heat transfer coefficient (U=W/m^2-K)
Ca,i = 1              % Feed Concentration (mol/m^3)
Ti = 350              % Feed Temperature (K)
Ca = 0.989            % Concentration of A in CSTR (mol/m^3)
T = 296.6             % Temperature in CSTR (K)

k = k0*exp(-EoverR/T)
rate = k * Ca
```



```
Parameters
  ! Manipulated Variables
  Tc = 270     ! Temperature of cooling jacket (K)

  ! Parameters
  q = 100      ! Volumetric Flowrate (m^3/sec)
  V = 100      ! Volume of CSTR (m^3)
  rho = 1000   ! Density of A-B Mixture (kg/m^3)
  Cp = .239    ! Heat capacity of A-B Mixture (J/kg-K)
  mdelH = 5e4  ! Heat of reaction for A->B (J/mol)
               ! E - Activation energy in the
               !  Arrhenius Equation (J/mol)
               ! R - Universal Gas Constant
               !  = 8.31451 J/mol-K
               ! EoverR = E/R
  EoverR = 8750
  k0 = 7.2e10  ! Pre-exponential factor (1/sec)
               ! U - Overall Heat Transfer
               !  Coefficient (W/m^2-K)
               ! A - Area - this value is specific
               !  for the U calculation (m^2)
               ! UA = U * A
  UA = 5e4
  Caf = 1      ! Feed Concentration (mol/m^3)
  Tf = 350     ! Feed Temperature (K)
End Parameters

Variables
  ! Differential States
  Ca = 0.9     ! Concentration of A in CSTR (mol/m^3)
  T  = 305     ! Temperature in CSTR (K)
End Variables

Equations
  ! note: the $ denotes time differential
  !  (e.g. $x is dx/dt)

  ! mole balance for species A
  V * $Ca = q*(Caf-Ca) - k0*V*exp(-EoverR/T)*Ca

  ! energy balance
  rho*Cp*V * $T=q*rho*Cp*(Tf-T) + V*mdelH*k0*exp(-EoverR/T) * Ca+UA*(Tc-T)
End Equations
```